# Maple Tips

## Chapter 12

## Lesson 12-1

### Approximating a Double Integral

The **ApproximateInt** command in the **Student[MultivariateCalculus]** package can be used approximate a double integral with several different methods.  The **output=plot** option will graph the function with its approximation prisms.

```
> restart: with(Student[MultivariateCalculus]):
> ApproximateInt(16-x^2-2*y^2, x=0..2, y=0..2, output = plot,
  partition = [2,2], method=lower);
```




```
> ApproximateInt(x-3*y^2, x=0..2, y=1..2, partition = [2,2]);
```

$$-11.87500000$$

# Lesson 12-2

## Calculating a Double Integral Over a Rectangular Region

The single-variable **int** command can be used recursively for iterated integrals over rectangular domains (**Int** can be used for notation purposes).

```
> restart:
```

```
> Int(Int(x^2*y, y=1..2), x=0..3) = int(int(x^2*y, y=1..2), x=0..3);
```

$$\int_0^3 \int_1^2 x^2\, y\, dy\, dx = \frac{27}{2}$$

# Lesson 12-3

## Calculating a Double Integral Over a General Region

The **int** command can be used recursively for a double integral over a general domain. The "inside" bounds must be dependent upon the "outside" variable.

```
> restart:
> Int(Int(x+2*y, y=2*x^2..1+x^2), x=-1..1) = int(int(x+2*y,
  y=2*x^2..1+x^2), x=-1..1);
```

$$\int_{-1}^{1} \int_{2x^2}^{1+x^2} x + 2\,y\, dy\, dx = \frac{32}{15}$$

# Lesson 12-4

## Calculating a Double Integral Over a Polar Region

Depending on the complexity of the iterated integral, the **int** command can be used to evaluate the integral as originally stated or after conversion to polar form.

```
> restart:
> Int(Int(1-x^2-y^2, y=-sqrt(1-x^2)..sqrt(1-x^2)), x=-1..1) =
  int(int(1-x^2-y^2, y=-sqrt(1-x^2)..sqrt(1-x^2)), x=-1..1);
```

$$\int_{-1}^{1} \int_{-\sqrt{1-x^2}}^{\sqrt{1-x^2}} 1 - x^2 - y^2\, dy\, dx = \frac{\pi}{2}$$

```
> Int(Int(r-r^3, r=0..1), theta=0..2*Pi) = int(int(r-r^3, r=0..1),
  theta=0..2*Pi);
```

$$\int_0^{2\pi} \int_0^1 r - r^3\, dr\, d\theta = \frac{\pi}{2}$$

# Lesson 12-6

## Calculating Surface Area

The **int** command can be used along with commands from the **VectorCalculus** package to calculate the area of a parametrically defined surface.

```
> restart: with(VectorCalculus):
```
Warning, the assigned names <,> and <|> now have a global binding

Warning, these protected names have been redefined and unprotected: *, +, -, ., D, Vector, diff, int, limit, series

```
> r:=(u,v)-><2*cos(u)*sin(v),2*sin(u)*sin(v),2*cos(v)>: r(u,v);
```
$$2\cos(u)\sin(v)\,\mathbf{e}_{\mathbf{x}} + 2\sin(u)\sin(v)\,\mathbf{e}_{\mathbf{y}} + 2\cos(v)\,\mathbf{e}_{\mathbf{z}}$$

```
> du:=diff(r(u,v),u); dv:=diff(r(u,v),v);
```
$$du := -2\sin(u)\sin(v)\,\mathbf{e}_{\mathbf{x}} + 2\cos(u)\sin(v)\,\mathbf{e}_{\mathbf{y}}$$

$$dv := 2\cos(u)\cos(v)\,\mathbf{e}_{\mathbf{x}} + 2\sin(u)\cos(v)\,\mathbf{e}_{\mathbf{y}} - 2\sin(v)\,\mathbf{e}_{\mathbf{z}}$$

```
> du &x dv;
```
$$-4\cos(u)\sin(v)^2\,\mathbf{e}_{\mathbf{x}} - 4\sin(u)\sin(v)^2\,\mathbf{e}_{\mathbf{y}} + \left(-4\sin(u)^2\sin(v)\cos(v) - 4\cos(u)^2\sin(v)\cos(v)\right)\mathbf{e}_{\mathbf{z}}$$

```
> Norm(du &x dv);
```
$$4\,\mathrm{csgn}(\sin(v))\sin(v)$$

```
> Int(Int(Norm(du &x dv), v=0..Pi), u=0..2*Pi) = int(int(Norm(du &x
  dv), v=0..Pi), u=0..2*Pi);
```
$$\int_0^{2\pi}\int_0^{\pi} 4\,\mathrm{csgn}(\sin(v))\sin(v)\,dv\,du = 16\pi$$

# Lesson 12-7

## Calculating a Triple Integral

Just as with double integrals, the **int** command can be used recursively for triple integrals over given domains regions.

```
> restart:
> Int(Int(Int(x^2*y*z^3, z=0..1), y=-2..1), x=-1..2) =
  int(int(int(x^2*y*z^3, z=0..1), y=-2..1), x=-1..2);
```
$$\int_{-1}^{2}\int_{-2}^{1}\int_{0}^{1} x^2\,y\,z^3\,dz\,dy\,dx = \frac{-9}{8}$$

```
> Int(Int(Int(z, z=0..1-x-y), y=0..1-x), x=0..1) = int(int(int(z,
  z=0..1-x-y), y=0..1-x), x=0..1);
```
$$\int_{0}^{1}\int_{0}^{1-x}\int_{0}^{1-x-y} z\,dz\,dy\,dx = \frac{1}{24}$$

# Lesson 12-8

## Triple Integrals in Cylindrical and Spherical Coordinates

Even more so than with double integrals in polar coordinates, evaluations are usually easier to calculate after converting to cylindrical or spherical coordinates (Maple struggles to evaluate the rectangular forms of the following examples).

```
> restart:
> (Int(Int(Int(x^2+y^2, z=sqrt(x^2+y^2)..2),
  y=-sqrt(4-x^2)..sqrt(4-x^2)), x=-2..2) = Int(Int(Int(r^3, z=r..2),
  r=0..2), theta=0..2*Pi)) = int(int(int(r^3, z=r..2), r=0..2),
  theta=0..2*Pi);
```

$$\left(\int_{-2}^{2}\int_{-\sqrt{4-x^2}}^{\sqrt{4-x^2}}\int_{\sqrt{x^2+y^2}}^{2} x^2+y^2\,dz\,dy\,dx = \int_{0}^{2\pi}\int_{0}^{2}\int_{r}^{2} r^3\,dz\,dr\,d\theta\right) = \frac{16\,\pi}{5}$$

```
> (Int(Int(Int(exp((x^2+y^2+z^2)^(3/2)),
  z=-sqrt(1-x^2-y^2)..sqrt(1-x^2-y^2)),
  y=-sqrt(1-x^2)..sqrt(1-x^2)), x=-1..1) =
  Int(Int(Int(exp(rho^3)*rho^2*sin(phi), rho=0..1), theta=0..2*Pi),
  phi=0..Pi)) = int(int(int(exp(rho^3)*rho^2*sin(phi), rho=0..1),
  theta=0..2*Pi), phi=0..Pi);
```

$$\left(\int_{-1}^{1}\int_{-\sqrt{1-x^2}}^{\sqrt{1-x^2}}\int_{-\sqrt{1-x^2-y^2}}^{\sqrt{1-x^2-y^2}} \mathbf{e}^{\left((x^2+y^2+z^2)^{(3/2)}\right)}\,dz\,dy\,dx = \int_{0}^{\pi}\int_{0}^{2\pi}\int_{0}^{1} \mathbf{e}^{(\rho^3)}\,\rho^2\sin(\phi)\,d\rho\,d\theta\,d\phi\right) =$$

$$-\frac{4\,\pi}{3}+\frac{4}{3}\,\mathbf{e}\,\pi$$

```
>
```